



3G company, INCORPORATED
RT 3, BOX 28A
GASTON, OREGON 97119

INSTRUCTIONS FOR THE USE OF THE 3G HORSE RACE HANDICAPPING PROGRAM

1. Fill out the data sheet for each race. (See "Instructions for filling out the data sheet")
2. LOAD the cassette provided
RUN the program.
3. The program will first ask you to enter the number of horses in the race. (maximum is 14) Then it will ask you to enter the data for the horse number 1.
4. At this point enter the numbers of the boxes checked on the data sheet. Note that the boxes are numbered 0 to 9. (The first box is 0 and not 1) For example, if box 0 (less than 7-days since last race) and box 3 (speed rating in top 3) were the only boxes checked, you would enter 03 (followed by the ENTER key)
Note that there is no comma or space between the numbers. If no boxes are checked for this horse, enter the letter N.
5. The program will then ask you to enter the data for each of the remaining horses in the race.
6. After you have entered the data for all the horses, the program will print out the data again, and you will have a chance to correct any mistakes.
7. Then the computer will calculate the odds for the race and print them out.
8. See "How to use the calculated odds".


INSTRUCTIONS FOR FILLING OUT THE DATA SHEET

The use of the program involves answering 10 questions about each horse. Through experience we have found that using a data sheet simplifies this task. A sample data sheet is enclosed. You should make copies of this data sheet - one for each race.

The data sheet has columns numbered 0 through 9, and rows numbered 1 through 14. Start with row 1 (for horse number 1), and if the answer to the question in column 0 is yes, check that box. If the answer is no, leave the box blank. Continue in this manner for all the columns. For example, if horse 1 has less than 7 days since it last raced, check box 0. If horse 1 does not have a speed rating in the top 3, leave box 3 blank.

To answer the questions, you will have to buy the "Daily Racing Form" the day before the race.

The following chart explains the meanings of the past performance figures in the "Racing Form". When you buy your "Racing Form" you should also pick up the free pamphlet: "How to Read Past Performances" for more detailed explanations.



The following are the 10 questions (0 through 9) that need to be answered for each horse in the race:

QUESTION 0:

Less than 7 days since last race for this horse?

The Day, Month, Year in the above chart indicated the date of the last race (field ①). If less than 7 days have elapsed since this horse raced, then check this box (Column 0) on the Data Sheet.


```
10 REM* HORSE RACING PROGRAM
20 REM* COPYRIGHT 1980 3G CO
30 REM* ALL RIGHTS RESERVED
35 CLS: PRINT"3G CO. HORSE RACE PREDICTION PROGRAM"
40 REM* READ IN STORED DATA
50 GOSUB 100
60 REM* READ IN VALUES FOR THIS RACE
65 GOSUB 200
70 REM* CALCULATE RESULTS
75 GOSUB 500
80 REM* PRINT OUT RESULTS
85 GOSUB 800
90 END
100 REM* THIS PART READS IN STORED DATA
110 FOR I=1 TO 10
120 READ PL(I)
130 NEXT I
140 REM* 1,7,9,13,14,17,18,20,21,23
150 DATA 1.814485,1.82577,1.60962,1.84781,1.45936
160 DATA 1.76827,1.53419,1.87386,1.4993,2.02993
170 RETURN
200 REM* READ IN VALUES FOR THIS RACE
210 DIM DA(10,15)
220 INPUT"ENTER THE NUMBER OF HORSES FOR THIS RACE";N
223 IF NE>14 PRINT"14 HORSES MAX.": GOTO 220
225 ME=NE: IS=1
230 FOR IH=IS TO ME
240 PRINT"ENTER DATA FOR HORSE ";IH
242 INPUT D1$
243 IF D1$="N" THEN D2(1)=-1: GOTO 252
244 FOR J=1 TO LEN(D1$)
246 D2(J)=VAL(MID$(D1$,J,1))
248 D2(J)=D2(J)+1
250 NEXT J
252 FOR I=1 TO 10
253 DA(I,IH)=0
254 FOR J=1 TO LEN(D1$)
256 IF I=D2(J) THEN DA(I,IH)=1 :GOTO 260
258 NEXT J
260 NEXT I
262 NEXT IH
264 PRINT"YOUR INPUT DATA LOOKS LIKE THIS:"
266 FOR IH=1 TO NE
268 PRINT "HORSE";IH,
269 FOR I=1 TO 10
270 IF DA(I,IH)=1 THEN PRINT I-1;
272 NEXT I:PRINT" "
274 NEXT IH
276 INPUT"DO YOU WANT TO CHANGE THE DATA";A$
280 IF A$="NO" THEN RETURN
290 INPUT"FOR WHICH HORSE";IH
300 ME=IH: IS=IH
310 GOTO 230
500 REM* THIS PART CALCULATES RESULTS
510 DIM PR(15),DS(15)
```



```
520 FOR I=1 TO 15:PR(I)=1:NEXT I
530 FOR IH=1 TO NE
540 FOR I= 1 TO 10
545 P1(I)=LOG(PL(I))
550 IF DA(I,IH)<>1 THEN GOTO 557
555 PP=P1(I): GOTO 560
557 IF P1(I)<0.5 THEN PP=1.-P1(I) ELSE PP=0.5
560 PR(IH)=PP*PR(IH)
570 NEXT I
580 NEXT IH
590 T1=0
600 FOR IH=1 TO NE
610 T1=T1+PR(IH)
620 NEXT IH
630 FOR IH=1 TO NE
640 PR(IH)=PR(IH)/T1
650 DS(IH)=1./PR(IH)-1.
660 NEXT IH
670 RETURN
800 REM* PRRINTS OUT RESULTS
810 CLS
820 PRINT"ODDS FOR THIS RACE:"
830 FOR IH=1 TO NE
840 PRINT "HORSE ";IH;" ODDS ";DS(IH)
850 NEXT IH
870 RETURN
```

*NOTE: IF YOU HAVE AN APPLE COMPUTER (OR ANY
COMPUTER WITHOUT AN 'ELSE' STATEMENT),*

REPLACE LINE 557 BY :

```
557 IF P1(I)<0.5 THEN PP=1.-P1(I): GO TO 560
558 PP=0.5
```


QUESTION 1:

Closing at finish or even in last race?

The small numbers or letter h (for head) in Field ① indicate how far behind the leader this horse was "In Stretch" and "At Finish". If, for example, the "In Stretch" small number is 3 (3 lengths behind the leader) and the "At Finish" small number is 2, then this horse is closing at finish, and you should check this box. (Always check this box if this horse won the last race)

QUESTION 2:

Class up more than \$2000 from last race.

Field ② indicates the "Claiming Price" for the last race. The number immediately above it in the past performance chart in your "Daily Racing form" is today's "Claiming Price". If today's claiming price is \$2000 more than the "Claiming Price" in the last race, then check this box on the data sheet. If there is not enough information about this factor - leave the box blank.

QUESTION 3:

Speed rating in top 3 in this race?

Check the speed rating (field ③) for all the horses in this race. If the speed rating for this horse is one of the 3 largest, then check this box. Use speed rating for the same distance as the race. Go back up to 3 races, if last race was not the same distance.

QUESTION 4:

Speed rating in low 3 in this race?

Check the speed rating for all the horses in this race. If the speed rating for this horse is one of the 3 lowest, then check this box.

QUESTION 5:

In money last race?

Look at the "At Finish" information. If the large number in the "At Finish" column is a 1, 2, or 3, then check this box.

QUESTION 6:

In last 3 in last race?

Again, look at the "At Finish" information. To know if the horse finished in the last 3 you will also have to look at field ⑥ - "Number of starters". Check this box if the horse finished in the last 3.

QUESTION 7:

Average winnings in top 3?

The winnings for this year are displayed right above the "Number of Starters" column in your "Daily Racing Form". If the winnings are among the 3 highest for horses in this race; check this box. Remember, we need the average winnings, so divide the total by the number of races (next to year) before comparing to other horses.

QUESTION 8:

Average winnings in low 3?

Check to see if the winnings for this year are among the 3 lowest. If so, then check this box.

QUESTION 9:

In top 3 favorites in THIS race?

Check your "Daily Racing Form" and look under the heading "Experts Selections". The last column is the "Consensus" column, where the 3 top horses picked by the "experts" are listed. If this horse is listed among these top 3 "Consensus" selections, then check this box.

HOW TO USE THE CALCULATED ODDS

Once you have the calculated odds for the race, it is important to apply them properly. It is generally not a good idea to bet on the horse with the lowest odds. Instead, you should bet on the horse for which the program has computed lower odds than the actual track odds.

For example, you go to the race track and shortly before the start of a race observe that horse number 1 has track odds 3:1 and horse number 2 has track odds 12:1. The previous evening the program has computed the odds on horse 1 as 3:1 and the odds on horse 2 as 5:1. You should then bet on horse number 2, since the program's odds are substantially lower than the track odds. Even if horse 2 only wins every 5 or 6 races, the payoff of 12:1 will make this kind of bet profitable in the long run.

DATA SHEET

NAME OF HORSE	NUMBER OF HORSE	LESS THAN 7 DAYS	CLOSING AT FINISH	CLASS UP #2000	SPEED TOP 3	SPEED LO 3	FIRST 3 IN LAST RACE	LAST 3 IN LAST RACE	WINNINGS HIGH 3	WINNINGS LOW 3	IN TOP 3 FAVORITES	NUMBERS OF THE COLUMNS CHECKED
		0	1	2	3	4	5	6	7	8	9	
	1											
	2											
	3											
	4											
	5											
	6											
	7											
	8											
	9											
	10											
	11											
	12											
	13											
	14											
		0	1	2	3	4	5	6	7	8	9	